

## F07BGF (SGBCON/DGBCON) – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

F07BGF (SGBCON/DGBCON) estimates the condition number of a real band matrix  $A$ , where  $A$  has been factorized by F07BDF (SGBTRF/DGBTRF).

### 2 Specification

```

SUBROUTINE F07BGF(NORM, N, KL, KU, AB, LDAB, IPIV, ANORM, RCOND,
1              WORK, IWORK, INFO)
ENTRY          sgbcon(NORM, N, KL, KU, AB, LDAB, IPIV, ANORM, RCOND,
1              WORK, IWORK, INFO)
INTEGER       N, KL, KU, LDAB, IPIV(*), IWORK(*), INFO
real         AB(LDAB,*), ANORM, RCOND, WORK(*)
CHARACTER*1   NORM

```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine estimates the condition number of a real band matrix  $A$ , in either the 1-norm or the infinity-norm:

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty.$$

Note that  $\kappa_\infty(A) = \kappa_1(A^T)$ .

Because the condition number is infinite if  $A$  is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine should be preceded by a call to F06RBF to compute  $\|A\|_1$  or  $\|A\|_\infty$ , and a call to F07BDF (SGBTRF/DGBTRF) to compute the  $LU$  factorization of  $A$ . The routine then uses Higham's implementation of Hager's method [1] to estimate  $\|A^{-1}\|_1$  or  $\|A^{-1}\|_\infty$ .

### 4 References

- [1] Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

### 5 Parameters

1: NORM — CHARACTER\*1 *Input*

*On entry:* indicates whether  $\kappa_1(A)$  or  $\kappa_\infty(A)$  is estimated as follows:

- if NORM = '1' or 'O', then  $\kappa_1(A)$  is estimated;
- if NORM = 'I', then  $\kappa_\infty(A)$  is estimated.

*Constraint:* NORM = '1', 'O' or 'I'.

2: N — INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

- 3:** KL — INTEGER *Input*  
*On entry:*  $k_l$ , the number of sub-diagonals within the band of  $A$ .  
*Constraint:*  $KL \geq 0$ .
- 4:** KU — INTEGER *Input*  
*On entry:*  $k_u$ , the number of super-diagonals within the band of  $A$ .  
*Constraint:*  $KU \geq 0$ .
- 5:** AB(LDAB,\*) — *real* array *Input*  
**Note:** the second dimension of the array AB must be at least  $\max(1,N)$ .  
*On entry:* the  $LU$  factorization of  $A$ , as returned by F07BDF (SGBTRF/DGBTRF).
- 6:** LDAB — INTEGER *Input*  
*On entry:* the first dimension of the array AB as declared in the (sub)program from which F07BGF (SGBCON/DGBCON) is called.  
*Constraint:*  $LDAB \geq 2 \times KL + KU + 1$ .
- 7:** IPIV(\*) — INTEGER array *Input*  
**Note:** the dimension of the array IPIV must be at least  $\max(1,N)$ .  
*On entry:* the pivot indices, as returned by F07BDF (SGBTRF/DGBTRF).
- 8:** ANORM — *real* *Input*  
*On entry:* if  $NORM = '1'$  or  $'O'$ , the 1-norm of the **original** matrix  $A$ ; if  $NORM = 'I'$ , the infinity-norm of the **original** matrix  $A$ . ANORM may be computed by calling F06RBF with the same value for the parameter NORM. ANORM must be computed either **before** calling F07BDF (SGBTRF/DGBTRF) or else from a **copy** of the original matrix  $A$ .  
*Constraint:*  $ANORM \geq 0.0$ .
- 9:** RCOND — *real* *Output*  
*On exit:* an estimate of the reciprocal of the condition number of  $A$ . RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than **machine precision**, then  $A$  is singular to working precision.
- 10:** WORK(\*) — *real* array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1,3*N)$ .
- 11:** IWORK(\*) — INTEGER array *Workspace*  
**Note:** the dimension of the array IWORK must be at least  $\max(1,N)$ .
- 12:** INFO — INTEGER *Output*  
*On exit:*  $INFO = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If  $INFO = -i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed estimate RCOND is never less than the true value  $\rho$ , and in practice is nearly always less than  $10\rho$ , although examples can be constructed where RCOND is much larger.

## 8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form  $Ax = b$  or  $A^T x = b$ ; the number is usually 4 or 5 and never more than 11. Each solution involves approximately  $2n(2k_l + k_u)$  floating-point operations (assuming  $n \gg k_l$  and  $n \gg k_u$ ) but takes considerably longer than a call to F07BEF (SGBTRS/DGBTRS) with 1 right-hand side, because extra care is taken to avoid overflow when  $A$  is approximately singular.

The complex analogue of this routine is F07BUF (CGBCON/ZGBCON).

## 9 Example

To estimate the condition number in the 1-norm of the matrix  $A$ , where

$$A = \begin{pmatrix} -0.23 & 2.54 & -3.66 & 0.00 \\ -6.98 & 2.46 & -2.73 & -2.13 \\ 0.00 & 2.56 & 2.46 & 4.07 \\ 0.00 & 0.00 & -4.78 & -3.82 \end{pmatrix}.$$

Here  $A$  is nonsymmetric and is treated as a band matrix, which must first be factorized by F07BDF (SGBTRF/DGBTRF). The true condition number in the 1-norm is 56.40.

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07BGF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
      INTEGER          NMAX, KLMAX, KUMAX, LDAB
      PARAMETER       (NMAX=8,KLMAX=8,KUMAX=8,LDAB=2*KLMAX+KUMAX+1)
      CHARACTER       NORM
      PARAMETER       (NORM='1')
*      .. Local Scalars ..
      real            ANORM, RCOND
      INTEGER          I, INFO, J, K, KL, KU, N
*      .. Local Arrays ..
      real            AB(LDAB,NMAX), WORK(3*NMAX)
      INTEGER          IPIV(NMAX), IWORK(NMAX)
*      .. External Functions ..
      real            F06RBF, X02AJF
      EXTERNAL         F06RBF, X02AJF
*      .. External Subroutines ..
      EXTERNAL         sgbcon, sgbtrf
*      .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07BGF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KL, KU
      IF (N.LE.NMAX .AND. KL.LE.KLMAX .AND. KU.LE.KUMAX) THEN
*
*          Read A from data file
*
          K = KL + KU + 1
```

```

      READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,N)
*
*      Compute norm of A
*
      ANORM = F06RBF(NORM,N,KL,KU,AB(KL+1,1),LDAB,WORK)
*
*      Factorize A
*
      CALL sgbtrf(N,N,KL,KU,AB,LDAB,IPIV,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*          Estimate condition number
*
          CALL sgbcon(NORM,N,KL,KU,AB,LDAB,IPIV,ANORM,RCOND,WORK,
+              IWORK,INFO)
*
          IF (RCOND.GE.X02AJF()) THEN
              WRITE (NOUT,99999) 'Estimate of condition number = ',
+              1.0e0/RCOND
          ELSE
              WRITE (NOUT,*) 'A is singular to working precision'
          END IF
          ELSE
              WRITE (NOUT,*) 'The factor U is singular'
          END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e10.2)
      END

```

## 9.2 Program Data

```

F07BGF Example Program Data
  4  1  2          :Values of N, KL and KU
-0.23  2.54 -3.66
-6.98  2.46 -2.73 -2.13
        2.56  2.46  4.07
        -4.78 -3.82 :End of matrix A

```

## 9.3 Program Results

F07BGF Example Program Results

Estimate of condition number = 5.64E+01